
'   PET hop sequence cross correlator
'   Checks the cross correlation between sequences

'   Pinnacle technologies
'   J.W. MacConnell
'   03-02-2000
'   Version 1.1
    ver$ = "1.1"

    DIM table%(255, 49), corr%(255), results%(255)

    CLS
    PRINT "Random  file name must be:  hoptable.txt"
    PRINT "Current file name must be:  chlist.txt"
    PRINT
    PRINT "Input File Type (Random or Current) (r or c) _ ";
    type$ = ""
    DO
        type$ = LCASE$(INKEY$)
        IF type$ <> "r" AND type$ <> "c" AND type$ <> "" THEN
            SOUND 700, 2
            type$ = ""
        END IF
    LOOP WHILE type$ = ""

    PRINT type$

    'handle Pinnacle and Infinetix file types
    IF type$ = "r" THEN
        'Pinnacle file
        OPEN "hoptable.txt" FOR INPUT AS #1

        'get rid of the Pinnacle header
        FOR nn% = 1 TO 3
            LINE INPUT #1, i$
        NEXT nn%
    ELSE
        'Infinetix file
        OPEN "chlist.txt" FOR INPUT AS #1
    END IF

    'read in the data to the table, and generate
    'a second copy of each sequence such that there
    'are two copies of the sequence lined up head to
    'tail.
    FOR seq% = 0 TO 255
        LINE INPUT #1, ii$
IF seq% = 0 THEN ix$ = ii$
        'get rid of the line header
        IF type$ = "r" THEN
            'Pinnacle line header
            position% = INSTR(1, ii$, ":")
        ELSE
            'Infinetix line header
            position% = INSTR(1, ii$, ",")
        END IF



        ii$ = MID$(ii$, position% + 1)

        'ii$ now contains the data.  It is the same
        'for either Pinnacle or Infinetix.

        'Put the data into the table
        FOR hop% = 0 TO 24
            'strip off a character
            IF hop% <> 24 THEN
                pointer% = INSTR(1, ii$, ",")
                hopval% = VAL(MID$(ii$, 1, pointer% - 1))

                'remove the data just obtained from the
                'string
                ii$ = MID$(ii$, pointer% + 1)
            ELSE
                hopval% = VAL(ii$)
            END IF

            table%(seq%, hop%) = hopval%
            table%(seq%, hop% + 25) = hopval%
        NEXT hop%
    NEXT seq%

    'Data is in the table, so close the input file
    CLOSE #1
'=========================================
    'do the correlation
    PRINT
    PRINT "Press any key to terminate"

    'Initialize the maximum correlation value
    'to the best case value
    totalcorrmax% = 0

    'Initialize the minimum correlation value
    'to the worst case value
    totalcorrmin% = 25

    'Initialize the mean correlation value
    'to the best case value
    totalcorrmean% = 0

    'the sequence being tested against all the
    'other sequences
    FOR testseq% = 0 TO 255

        'Allow an exit method
        IF INKEY$ <> "" THEN END

        'The maximum value of the correlation that
        'this test sequence has against any of the
        'other sequences, except itself, which is
        'always 25
        maxcorr% = 0

        LOCATE 8, 5



        PRINT SPACE$(50)
        LOCATE 8, 5
        PRINT "Working on Test Sequence # "; LTRIM$(RTRIM$(STR$(testseq%)))

        'The sequence it is being tested against
            FOR seq% = 0 TO 255

             'don't test the test sequence against itself
             'since this will always yield 25
              IF seq% <> testseq% THEN

                FOR lag% = 0 TO 24
                    corr% = 0
                    'Does the correlation for a given lag%
                    FOR nn% = 0 TO 24
                        IF table%(testseq%, nn%) = table%(seq%, nn% + lag%) THEN
                            corr% = corr% + 1
                        END IF
                    NEXT nn%

                    'see if this lag had a larger correlation that
                    'the previous lag.  If so, save the large value
                    IF corr% > maxcorr% THEN
                        maxcorr% = corr%
                    END IF
                NEXT lag%
              END IF
            NEXT seq%

            results%(testseq%) = maxcorr%

            'Find the maximum correlation for all
            'the different sequences
            IF maxcorr% > totalcorrmax% THEN
                totalcorrmax% = maxcorr%
            END IF

            'Find the minimum correlation for all
            'the different sequences
            IF maxcorr% < totalcorrmin% THEN
                totalcorrmin% = maxcorr%
            END IF

            'Compute the mean correlation for all
            'the different sequences
            totalcorrmean% = totalcorrmean% + maxcorr%

    NEXT testseq%

    'All done with the correlation, now print the results
    totalcorrmean = INT(totalcorrmean% / 256 * 100) / 100
    PRINT
    PRINT "The Maximum Correlation was "; totalcorrmax%
    PRINT "The Minimum Correlation was "; totalcorrmin%
    PRINT "The Mean    Correlation was "; totalcorrmean
    PRINT



    OPEN "hop_corr.txt" FOR OUTPUT AS #1
    PRINT
    PRINT "Printing results to the file HOP_CORR.TXT"

    PRINT #1, "Sequence types = ";
    IF type$ = "r" THEN
        PRINT #1, "Random"
    ELSE
        PRINT #1, "Current"
    END IF

    PRINT #1,
    PRINT #1, "The Maximum Correlation was "; totalcorrmax%
    PRINT #1, "The Minimum Correlation was "; totalcorrmin%
    PRINT #1, "The Mean    Correlation was "; totalcorrmean

    PRINT #1,

    FOR seq% = 0 TO 255
        PRINT #1, seq%, results%(seq%)

    NEXT seq%

    CLOSE 1
    PRINT
    PRINT "All Done"

    END


